HP-35s Calculator Program — Closure 3AE

Three-Point Horizontal Resection Reduction Program

Programmer: Dr. Bill Hazelton — **European Edition, using Grads** (gon)

Date: June, 2013. **Version**: 1.0 **Mnemonic**: R for Resection

Line	Instruction	Program Entry Keystrokes
R001	LBL R	LBL R
R002	CLSTK	CLEAR 5 STK
R003	GRAD	MODE 3 GRD
R004	FS? 10	FLAGS 3 FS? . 0
R005	GTO R009	GTO R 009
R006	SF 1	FLAGS 1 SF 1
R007	SF 10	FLAGS 1 SF . 0
R008	GTO R010	GTO R 010
R009	CF 1	FLAGS 2 CF 1
R010	RESECTION	(Key in using EQN RCL R RCL E RCL S etc.)
R011	PSE	PSE
R012	ENTER LEFT X	(Key in using EQN RCL E RCL N RCL T etc.)
R013	PSE	PSE
R014	INPUT X	INPUT X
R015	STO A	STO A
R016	ENTER LEFT Y	(Key in using EQN RCL E RCL N RCL T etc.)
R017	PSE	PSE
R018	INPUT Y	INPUT Y
R019	STO B	STO B
R020	ENTER MID X	(Key in using EQN RCL E RCL N RCL T etc.)
R021	PSE	PSE
R022	INPUT X	INPUT X
R023	STO C	STO C
R024	ENTER MID Y	(Key in using EQN RCL E RCL N RCL T etc.)
R025	PSE	PSE
R026	INPUT Y	INPUT Y
R027	STO D	STO D
R028	ENTER RIGHT X	(Key in using EQN RCL E RCL N RCL T etc.)
R029	PSE	PSE
R030	INPUT X	INPUT X
R031	STO E	STO E
R032	ENTER RIGHT Y	(Key in using EQN RCL E RCL N RCL T etc.)
R033	PSE	PSE
R034	INPUT Y	INPUT Y
R035	STO F	STO F
R036 **	ENTER ALPHA	(Key in using EQN RCL E RCL N RCL T etc.) **
R037	PSE	PSE
R038	INPUT X	INPUT X
R039	STO G	STO G

Three Point Horizontal Resection Reduction Program

R040	ENTER BETA	(Key in using EQN RCL E RCL N RCL T etc.)
R041	PSE	PSE
R042	INPUT X	INPUT X
R043	STO H	STO H
R044	RCL B	RCL B
R045	RCL– D	RCL – D
R046	RCL A	RCL A
R047	RCL– C	RCL – C
R048	0 **i** 1	Press the zero key, then **i**, then 1.
R049	×	
R050	+	
R051	STO L	STO L
R052	RCL F	RCL F
R053	RCL– D	RCL – D
R054	RCL E	RCL E
R055	RCL– C	RCL – C
R056	0 **i** 1	Press the zero key, then **i**, then 1.
R057	×	
R058	+	
R059	STO K	STO K
R060	400	
R061	STO Z	STO Z
R062	RCL L	RCL L
R063	ARG	ARG
R064	RCL K	RCL K
R065	ARG	ARG
R066	–	
R067	x < 0 ?	x?0 3 <
R068	RCL+ Z	RCL + Z
R069	STO I	STO I
R070	RCL+ G	RCL + G
R071	RCL+ H	RCL + H
R072	RCL Z	RCL Z
R073	x < > y	
R074	–	
R075	STO S	STO S
R076	RCL L	RCL L
R077	ABS	ABS
R078	RCL H	RCL H
R079	SIN	
R080	×	
R081	RCL K	RCL K
R082	ABS	ABS
R083	÷	(Division /)
R084	RCL G	RCL G
R085	SIN	
R086	÷	(Division /)

Three Point Horizontal Resection Reduction Program

R087	RCL S	RCL **S**
R088	SIN	
R089	÷	(Division /)
R090	RCL S	RCL **S**
R091	TAN	
R092	1/x	
R093	+	
R094	1/x	
R095	ATAN	**ATAN**
R096	STO X	**STO X**
R097	RCL L	RCL **S**
R098	ARG	**ARG**
R099	200	
R100	+	
R101	RCL+ G	RCL + **G**
R102	RCL+ X	RCL + **X**
R103	STO Y	**STO Y**
R104	RCL L	RCL **L**
R105	ABS	**ABS**
R106	RCL X	RCL **X**
R107	SIN	
R108	×	
R109	RCL G	RCL **G**
R110	SIN	
R111	÷	(Division /)
R112	STO J	**STO J**
R113	RCL Y	RCL **Y**
R114	SIN	
R115	×	
R116	STO P	**STO P**
R117	RCL J	RCL **J**
R118	RCL Y	RCL **Y**
R119	COS	
R120	×	
R121	STO Q	**STO Q**
R122	RCL P	RCL **P**
R123	RCL+ C	RCL + **C**
R124	STO X	**STO X**
R125	UNKNOWN X =	(Key in as EQN RCL U RCL N RCL K etc.)
R126	PSE	**PSE**
R127	VIEW X	**VIEW X**
R128	UNKNOWN Y =	(Key in as EQN RCL U RCL N RCL K etc.)
R129	PSE	**PSE**
R130	RCL Q	RCL **Q**
R131	RCL+ D	RCL + **D**
R132	STO Y	**STO Y**
R133	VIEW Y	**VIEW Y**

Three Point Horizontal Resection Reduction Program

R134	RCL I	RCL **I**
R135	RCL+ G	RCL + **G**
R136	RCL+ H	RCL + **H**
R137	CHECK VALUE =	(Key in using EQN RCL C RCL H RCL E etc.)
R138	PSE	**PSE**
R139	STOP	(Key in R/S)
R140	FS? 1	**FLAGS** 3 FS? 1
R141	CF 10	**FLAGS** 2 CF .0
R142	RTN	**RTN**

Notes

(1) Horizontal 3-point resection solution, based on measuring two angles between three known points at an unknown point, the location of which is to be computed.

(2) Brief prompts are provided before each requirement for data entry, as well as before results are displayed. Each prompt shows for about 1 second, and is then replaced by the value or request for input.

(3) Co-ordinates of the unknown point are displayed following brief prompts. They are also stored in storage registers X and Y for later retrieval.

(4) Angles are entered and displayed in grads (gon). Internal storage of lines uses the calculator's complex number format.

Theory

This 2-D resection uses Ormsby's solution. In the discussion below, A is the left point, B is the middle point, C is the right point, and P is the unknown point. The left angle is alpha (α) and the right angle is beta (β). The interior angle at B is gamma (γ). The angle at point A is x, which is the first objective of the solution. A diagram is shown on the next page.

α and β are angles observed from the point P to points A, B and C, whose co-ordinate are known.

$$BP = \frac{AB \sin x}{\sin \alpha} = \frac{BC \sin y}{\sin \beta}$$

$$\text{and } (x + y) = (360° - (\alpha + \beta + \gamma)) = s$$

$$\frac{AB}{\sin \alpha} \sin x = \frac{BC}{\sin \beta} \sin (s - x) = \frac{BC}{\sin \beta} (\sin s \cos x - \cos s \sin x)$$

$$\frac{AB}{\sin \alpha} \sin x = \frac{BC}{\sin \beta} \sin s \cos x - \frac{BC}{\sin \beta} \cos s \sin x$$

$$\sin x \left(\frac{AB}{\sin \alpha} + \frac{BC}{\sin \beta} \cos s\right) = \frac{BC}{\sin \beta} \sin s \cos x$$

$$\left(\frac{AB}{\sin \alpha} + \frac{BC}{\sin \beta} \cos s\right) \frac{\sin \beta}{BC \sin s} = \cot x$$

Three Point Horizontal Resection Reduction Program

$$\frac{AB \sin\beta}{BC \sin\alpha \sin s} + \frac{BC \cos s \sin\beta}{BC \sin s \sin\beta} = \cot x$$

$$\frac{AB \sin\beta}{BC \sin\alpha \sin s} + \cot s = \cot x \qquad \text{[this is the equation solved first]}$$

$$y = s - x$$

With x and y determined, the sides AP, BP and CP can be calculated and hence the co-ordinates of P, as follows:

The azimuth of BP (Az_{BP}) can be determined using $Az_{BP} = Az_{AB} + \alpha + x$

The length of BP can be determined using $BP = \frac{AB \sin x}{\sin\alpha}$



Knowing the co-ordinates of B, Az_{BP} and BP, the co-ordinates of P can be easily computed. As a check, the equivalent solution can be obtain through the sides AP or CP, or using the angle y. Note that if P is close the danger circle, a solution will still be obtained, but the sum of $\alpha + \beta + \gamma$ will be close to 200^g, probably in the range 195^g to 205^g. In this case, the solution will be highly sensitive to changes in α and β. If the solution is close to the danger circle, recompute with the angles changed by about their precision and see how much the resulting co-ordinates change. It can be quite surprising! To facilitate this, press GTO R036, then R/S, to run the program with the same known points, but you can enter different observed angles.

Azimuths in grads (gon) are used. Arbitrary co-ordinates are satisfactory. Plane surveying assumptions apply. The program uses no error checking on entered data. A check is made by showing the sum $\alpha + \beta + \gamma$. If this is close to 200^g, the unknown point lies close to the danger circle and the result is highly suspect.

Three Point Horizontal Resection Reduction Program

Sample Computation 1

Known Points

Point Name	X	Y
Point A	–25.336	778.136
Point B	–27.465	1179.927
Point C	–30.297	1555.643

Angles Left (α) = 136.3526 grads

Right (β) = 27.1924 grads

Results Unknown Point (P) X Co-ordinate = 75.702

Unknown Point (P) Y Co-ordinate = 1038.192

Check Angle = 363.6875 grads

Sample Computation 2

Known Points

Point Name	X	Y
Point A	133.639	1548.712
Point B	158.065	1492.276
Point C	150.267	1353.056

Angles Left (α) = 5.0148 grads

Right (β) = 3.4129 grads

Results Unknown Point (P) X Co-ordinate = 109.509

Unknown Point (P) Y Co-ordinate = 1143.181

Check Angle = 178.8618 grads

This is not the ideal arrangement for a resection, as the measured angles are quite small. But the program will still produce an acceptable result.

This example is provided because the other example has negative co-ordinates and this tends to increase the chances of incorrect data entry. It happened to me, twice!

Three Point Horizontal Resection Reduction Program

Running the Program

Press XEQ R ENTER

Calculator displays RESECTION briefly, so that you know you are running the correct program.

Prompt ENTER LEFT X briefly, then X?

Enter X Co-ordinate for left known point.

Press R/S.

Prompt ENTER LEFT Y briefly, then Y?

Enter Y Co-ordinate for left known point.

Press R/S.

Prompt ENTER MID X briefly, then X?

Enter X Co-ordinate for middle known point.

Press R/S.

Prompt ENTER MID Y briefly, then Y?

Enter Y Co-ordinate for middle known point.

Press R/S.

Prompt ENTER RIGHT X briefly, then X?

Enter X Co-ordinate for right known point.

Press R/S.

Prompt ENTER RIGHT Y briefly, then Y?

Enter Y Co-ordinate for right known point.

Press R/S.

** Prompt ENTER ALPHA briefly, then X?

Enter left angle (α) in grads (gon).

Press R/S.

Prompt ENTER BETA briefly, then X?

Enter right angle (β) in grads (gon).

Press R/S.

Calculator displays RUNNING while doing the calculations.

Prompt UNKNOWN X briefly, then X=

X co-ordinate of unknown point (P) is displayed.

Press R/S.

Prompt UNKNOWN Y briefly, then Y=

Y co-ordinate of unknown point (P) is displayed.

Press R/S.

Prompt CHECK VALUE briefly.

Sum $\alpha + \beta + \gamma$ is displayed in lower line of display in grads (gon).

Check that value is not too close to 200^g. At least 5^g away, preferably 15^g or more away.

If you want to re-run the program with the same fixed points but different angles, press GTO R036, then R/S, and the program will start from the step labeled ** above, prompting with ENTER ALPHA. Changing the angles by small amounts can give you a good idea of the reliability of the solution.

Once you are finished with the program, and you have seen the CHECK VALUE prompt and value, press R/S to clear flags. The program ends.

Storage Registers Used

A Left known point – X co-ordinate

B Left known point – Y co-ordinate

C Middle known point – X co-ordinate

D Middle known point – Y co-ordinate

E Right known point – X co-ordinate

F Right known point – Y co-ordinate

G Left measured angle — alpha (α)

H Right measured angle — beta (β)

I Interior angle at Middle known point — gamma (γ)

J Distance from middle point to the unknown point

K Vector from middle to right point (complex number format)

L Vector from middle to left point (complex number format)

P X co-ordinate of unknown point

Q Y co-ordinate of unknown point

S s = x + y in grads

X Initial inputs, then angle x, then X co-ordinate of unknown point

Y Initial inputs, then azimuth from middle to unknown point, then Y co-ordinate of unknown point

Z 400

Labels Used

Label **R** Length = 575 Checksum = 7082

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. Use the sample computation to check proper operation after entry.

The program sets flag 10, to allow equations to be displayed as prompts, and at the end of the program, resets flag 10 to its previous setting. The program uses flag 1 to record the state of flag 10 before the program started.

Note that if you change anything in the program, especially prompts, the Checksum and Length values will probably change.

Localization

You are very welcome to modify this program to suit your local circumstances. The easiest step is changing the prompts to local language or terminology. You may prefer to use E and N for the co-ordinates, and some other names for the angles.

If you want to change the variable prompts, this will require more changes within the program. As many of the variables (and letters) are used, this may cause you to have to change other variables. For example, if you want to use Z for angular input, you will then need to use a different variable to store the 400 value used for correcting negative angles. Make sure you have tracked all the impacts of your changes before you start implementing them!

This version was specifically designed for European use, based on the use of grads (gon) for angles. Another version (Closure 3A) is available for degrees, minutes and seconds use. If you want to use decimal degrees throughout, change Line R003 to: MODE 2 DEG and use decimal degrees in the program. You will also need to make the following changes:

Line R060 to 360

Line R099 to 180